

# The Stochastic Power Law Logistic Model: Derivations of Asymptotic Approximations of the first 3 Cumulants of the QSD, with $s=7$ .

Ingemar Nåsell

The procedure **cumf** is used to determine the cumulant of order  $kk$  as a function of the (raw) moments of orders up to  $kk$ , while the procedure **muf** determines the (raw) moment of order  $kk$  as a function of the cumulants of orders up to  $kk$ . The procedure **ODEcum3** derives ODEs for the first 3 cumulants.

More details are given in the Maple work-sheet DeriveODECum3.

```
> restart;
> cumf:=proc(kk)
  local M,K,Ks,cum;
  description "Determines the cumulant of order kk as function of
the raw moments of orders up to kk";
  M:=1+add(cat(mu,k)*theta^k/k!,k=1..kk);
  K:=log(M);
  Ks:=convert(series(K,theta,kk+1),polynom);
  cum:=sort(simplify(coeff(Ks,theta,kk)*kk!),[seq(cat(mu,j),j=1..
kk)],plex);
end proc;

> muf:=proc(kk)
  local K,M,Ms,mu;
  description "Determines the (raw) moment of order kk as function
of the cumulants of orders up to kk";
  K:=add(cat(kappa,k)*theta^k/k!,k=1..kk);
  M:=exp(K);
  Ms:=convert(series(M,theta,kk+1),polynom);
  mu:=sort(coeff(Ms,theta,kk)*kk!,[seq(cat(kappa,j),j=1..kk)],
plex);
end proc;

> ODEcum3:=proc(s)
  local Dmu1,Dmu2,Dmu3,sub1,sub2,Dcum1,Dcum2,Dcum3,a,c,d,b;
  description "Derive ODEs for the first 3 cumulants";
  Dmu1:=a*mu1-b*cat(mu,s+1);
  Dmu2:=2*a*mu2+c*mu1-d*cat(mu,s+1)-2*b*cat(mu,s+2);
  Dmu3:=a*(mu1+3*mu3)+3*c*mu2-3*d*cat(mu,s+2)-b*(cat(mu,s+1)+3*
cat(mu,s+3));
  sub1:=D(mu1)=Dmu1,D(mu2)=Dmu2,D(mu3)=Dmu3;
  sub2:=seq(cat(mu,i)=muf(i),i=1..3+s);
  Dcum1:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
(1))))),[a,c,d,b])));
```

```

Dcum2:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
(2))))),[a,c,d,b]));
Dcum3:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
(3))))),[a,c,d,b]));
a:=mu*(R0-1);
c:=mu*(R0+1);
d:=mu*(R0-alpha)/N^s;
b:=mu*(R0+alpha)/N^s;
[eval(Dcum1),eval(Dcum2),eval(Dcum3)];
end proc:

```

Put s=7 and denote the derivatives of the first 3 cumulants by A, B, C.

```
> s:=7;
```

```
s := 7
```

(1)

```
> A:=ODEcum3(s)[1];
B:=ODEcum3(s)[2];
C:=ODEcum3(s)[3];
```

$$\begin{aligned}
A := & \mu (R0 - 1) \kappa 1 - \frac{1}{N^7} \left( (\kappa 1^8 + 28 \kappa 1^6 \kappa 2 + 56 \kappa 1^5 \kappa 3 + (210 \kappa 2^2 + 70 \kappa 4) \kappa 1^4 \right. \\
& + 105 \kappa 2^4 + (560 \kappa 2 \kappa 3 + 56 \kappa 5) \kappa 1^3 + 210 \kappa 2^2 \kappa 4 + (420 \kappa 2^3 + 420 \kappa 2 \kappa 4 \\
& + 280 \kappa 3^2 + 28 \kappa 6) \kappa 1^2 + 56 \kappa 3 \kappa 5 + 35 \kappa 4^2 + (840 \kappa 2^2 \kappa 3 + 168 \kappa 2 \kappa 5 \\
& \left. + 280 \kappa 3 \kappa 4 + 8 \kappa 7) \kappa 1 + (280 \kappa 3^2 + 28 \kappa 6) \kappa 2 + \kappa 8) \mu (R0 + \alpha) \right)
\end{aligned}$$

$$\begin{aligned}
B := & 2 \mu (R0 - 1) \kappa 2 + \mu (R0 + 1) \kappa 1 - \frac{1}{N^7} \left( (\kappa 1^8 + 28 \kappa 1^6 \kappa 2 + 56 \kappa 1^5 \kappa 3 \right. \\
& + (210 \kappa 2^2 + 70 \kappa 4) \kappa 1^4 + 105 \kappa 2^4 + (560 \kappa 2 \kappa 3 + 56 \kappa 5) \kappa 1^3 + 210 \kappa 2^2 \kappa 4 \\
& + (420 \kappa 2^3 + 420 \kappa 2 \kappa 4 + 280 \kappa 3^2 + 28 \kappa 6) \kappa 1^2 + 56 \kappa 3 \kappa 5 + 35 \kappa 4^2 \\
& + (840 \kappa 2^2 \kappa 3 + 168 \kappa 2 \kappa 5 + 280 \kappa 3 \kappa 4 + 8 \kappa 7) \kappa 1 + (280 \kappa 3^2 + 28 \kappa 6) \kappa 2 \\
& \left. + \kappa 8) \mu (R0 - \alpha) \right) - \frac{1}{N^7} \left( 2 (8 \kappa 1^7 \kappa 2 + 28 \kappa 1^6 \kappa 3 + (168 \kappa 2^2 + 56 \kappa 4) \kappa 1^5 \right. \\
& + (700 \kappa 2 \kappa 3 + 70 \kappa 5) \kappa 1^4 + 1260 \kappa 2^3 \kappa 3 + (840 \kappa 2^3 + 840 \kappa 2 \kappa 4 + 560 \kappa 3^2 \\
& + 56 \kappa 6) \kappa 1^3 + 378 \kappa 2^2 \kappa 5 + 280 \kappa 3^3 + (2940 \kappa 2^2 \kappa 3 + 588 \kappa 2 \kappa 5 + 980 \kappa 3 \kappa 4 \\
& + 28 \kappa 7) \kappa 1^2 + 84 \kappa 3 \kappa 6 + 126 \kappa 4 \kappa 5 + (840 \kappa 2^4 + 1680 \kappa 2^2 \kappa 4 + 448 \kappa 3 \kappa 5 \\
& + 280 \kappa 4^2 + (2240 \kappa 3^2 + 224 \kappa 6) \kappa 2 + 8 \kappa 8) \kappa 1 + (1260 \kappa 3 \kappa 4 + 36 \kappa 7) \kappa 2 \\
& \left. + \kappa 9) \mu (R0 + \alpha) \right)
\end{aligned}$$

$$\begin{aligned}
C := & (\kappa 1 + 3 \kappa 3) \mu (R0 - 1) + 3 \mu (R0 + 1) \kappa 2 - \frac{1}{N^7} \left( 3 (8 \kappa 1^7 \kappa 2 + 28 \kappa 1^6 \kappa 3 \right. \\
& + (168 \kappa 2^2 + 56 \kappa 4) \kappa 1^5 + (700 \kappa 2 \kappa 3 + 70 \kappa 5) \kappa 1^4 + 1260 \kappa 2^3 \kappa 3 + (840 \kappa 2^3 \\
& + 840 \kappa 2 \kappa 4 + 560 \kappa 3^2 + 56 \kappa 6) \kappa 1^3 + 378 \kappa 2^2 \kappa 5 + 280 \kappa 3^3 + (2940 \kappa 2^2 \kappa 3 \\
& \left. + 588 \kappa 2 \kappa 5 + 980 \kappa 3 \kappa 4 + 28 \kappa 7) \kappa 1^2 + 84 \kappa 3 \kappa 6 + 126 \kappa 4 \kappa 5 + (840 \kappa 2^4 \right.
\end{aligned}$$

(2)

$$\begin{aligned}
& + 1680 \kappa_2^2 \kappa_4 + 448 \kappa_3 \kappa_5 + 280 \kappa_4^2 + (2240 \kappa_3^2 + 224 \kappa_6) \kappa_2 + 8 \kappa_8) \kappa_1 \\
& + (1260 \kappa_3 \kappa_4 + 36 \kappa_7) \kappa_2 + \kappa_9) \mu (R_0 - \alpha) - \frac{1}{N^7} \left( 24 \left( \frac{\kappa_1^8}{24} + \kappa_1^7 \kappa_3 \right. \right. \\
& + \left( 7 \kappa_2^2 + \frac{7}{6} \kappa_2 + \frac{7}{2} \kappa_4 \right) \kappa_1^6 + \left( 63 \kappa_2 \kappa_3 + \frac{7}{3} \kappa_3 + 7 \kappa_5 \right) \kappa_1^5 + 105 \kappa_2^5 \\
& + \left( 105 \kappa_2^3 + \frac{35}{4} \kappa_2^2 + \frac{245}{2} \kappa_2 \kappa_4 + \frac{175}{2} \kappa_3^2 + \frac{35}{12} \kappa_4 + \frac{35}{4} \kappa_6 \right) \kappa_1^4 \\
& + \frac{35 \kappa_2^4}{8} + \frac{735 \kappa_2^3 \kappa_4}{2} + \left( 665 \kappa_2^2 \kappa_3 + 245 \kappa_3 \kappa_4 + \left( \frac{70 \kappa_3}{3} + 140 \kappa_5 \right) \kappa_2 \right. \\
& + \left. \frac{7 \kappa_5}{3} + 7 \kappa_7 \right) \kappa_1^3 + \frac{525 \kappa_3^2 \kappa_4}{2} + \left( 315 \kappa_2^4 + \frac{35 \kappa_2^3}{2} + \frac{1365 \kappa_2^2 \kappa_4}{2} \right. \\
& + \left. \frac{35 \kappa_3^2}{3} + 196 \kappa_3 \kappa_5 + \frac{245 \kappa_4^2}{2} + \left( 945 \kappa_3^2 + \frac{35 \kappa_4}{2} + \frac{189 \kappa_6}{2} \right) \kappa_2 + \frac{7 \kappa_6}{6} \right. \\
& + \left. \frac{7 \kappa_8}{2} \right) \kappa_1^2 + \left( \frac{1505 \kappa_3^2}{2} + \frac{35 \kappa_4}{4} + \frac{301 \kappa_6}{4} \right) \kappa_2^2 + \frac{35 \kappa_4^2}{24} + \frac{105 \kappa_4 \kappa_6}{4} \\
& + \frac{63 \kappa_5^2}{4} + \left( 1155 \kappa_2^3 \kappa_3 + 280 \kappa_3^3 + (35 \kappa_3 + 357 \kappa_5) \kappa_2^2 + 126 \kappa_4 \kappa_5 \right. \\
& + (1225 \kappa_3 \kappa_4 + 7 \kappa_5 + 35 \kappa_7) \kappa_2 + \left( \frac{35 \kappa_4}{3} + 84 \kappa_6 \right) \kappa_3 + \frac{\kappa_7}{3} + \kappa_9) \kappa_1 \\
& + \frac{\kappa_{10}}{8} + \left( \frac{35}{3} \kappa_3^2 + 308 \kappa_3 \kappa_5 + \frac{385}{2} \kappa_4^2 + \frac{7}{6} \kappa_6 + \frac{11}{2} \kappa_8 \right) \kappa_2 + \left( \frac{7 \kappa_5}{3} \right. \\
& \left. + 15 \kappa_7 \right) \kappa_3 + \frac{\kappa_8}{24} \Big) \mu (R_0 + \alpha) \Big)
\end{aligned}$$

Determine asymptotic approximations of the first 3 cumulants of the following forms. Assume also that  $\kappa_4 - \kappa_{10}$  are  $O(N)$ .

```

> kappa1:=x1*N + x2 + x3/N:
kappa2:=y1*N + y2:
kappa3:=z1*N:
kappa4:=u1*N:
kappa5:=u2*N:
kappa6:=u3*N:
kappa7:=u4*N:
kappa8:=u5*N:
kappa9:=u6*N:
kappa10:=u7*N:

```

Asymptotic expressions for A, B, C are written  $A = A_1*N + A_2 + A_3/N$ ,  $B = B_1*N + B_2$ ,  $C = C_1*N$ , where

```

> A1:=coeff(A,N,1); A2:=coeff(A,N,0); A3:=coeff(A,N,-1);
B1:=coeff(B,N,1); B2:=coeff(B,N,0);
C1:=coeff(C,N,1);

```

$$\begin{aligned}
A1 &:= \mu (R0 - 1) x1 - x1^8 \mu (R0 + \alpha) \\
A2 &:= \mu (R0 - 1) x2 - (8 x2 x1^7 + 28 x1^6 y1) \mu (R0 + \alpha) \\
A3 &:= \mu (R0 - 1) x3 - (2 (2 (2 x3 x1 + x2^2) x1^2 + 4 x2^2 x1^2) x1^4 + 16 x2^2 x1^6 \\
&\quad + 28 x1^6 y2 + 168 x2 x1^5 y1 + 56 x1^5 z1 + 210 y1^2 x1^4) \mu (R0 + \alpha) \\
B1 &:= 2 \mu (R0 - 1) y1 + \mu (R0 + 1) x1 - x1^8 \mu (R0 - \alpha) - 16 x1^7 y1 \mu (R0 + \alpha) \\
B2 &:= 2 \mu (R0 - 1) y2 + \mu (R0 + 1) x2 - (8 x2 x1^7 + 28 x1^6 y1) \mu (R0 - \alpha) \\
&\quad - 2 (8 x1^7 y2 + 56 x2 x1^6 y1 + 28 x1^6 z1 + 168 y1^2 x1^5) \mu (R0 + \alpha) \\
C1 &:= (x1 + 3 z1) \mu (R0 - 1) + 3 \mu (R0 + 1) y1 - 24 x1^7 y1 \mu (R0 - \alpha) \\
&\quad - 24 \left( \frac{1}{24} x1^8 + x1^7 z1 + 7 y1^2 x1^6 \right) \mu (R0 + \alpha)
\end{aligned} \tag{3}$$

6 equations can now be formed by setting each of A1, A2, A3, B1, B2, C1 equal to zero. These equations can be solved for the 6 unknowns x1, x2, x3, y1, y2, z1. The solutions are found sequentially, as follows:

First we solve A1=0 for x1. This equation has s+1=8 solutions. Among them, we exclude s=7 spurious solution.

After this, B1=0 is solved for y1, A2=0 is solved for x2, C1=0 is solved for z1, B2=0 is solved for y2, and A3=0 is solved for x3.

**> x1a:=solve(A1,x1);**

**x1:=((R0-1)/(R0+alpha))^(1/7);**

$$x1a := 0, \frac{((R0-1)(R0+\alpha)^6)^{1/7}}{R0+\alpha},$$

$$\frac{\left( \cos\left(\frac{2\pi}{7}\right) + I \cos\left(\frac{3\pi}{14}\right) \right) ((R0-1)(R0+\alpha)^6)^{1/7}}{R0+\alpha},$$

$$\frac{\left( -\cos\left(\frac{3\pi}{7}\right) + I \cos\left(\frac{\pi}{14}\right) \right) ((R0-1)(R0+\alpha)^6)^{1/7}}{R0+\alpha},$$

$$\frac{\left( -\cos\left(\frac{\pi}{7}\right) + I \cos\left(\frac{5\pi}{14}\right) \right) ((R0-1)(R0+\alpha)^6)^{1/7}}{R0+\alpha},$$

$$\frac{\left( -\cos\left(\frac{\pi}{7}\right) - I \cos\left(\frac{5\pi}{14}\right) \right) ((R0-1)(R0+\alpha)^6)^{1/7}}{R0+\alpha},$$

$$\frac{\left( -\cos\left(\frac{3\pi}{7}\right) - I \cos\left(\frac{\pi}{14}\right) \right) ((R0-1)(R0+\alpha)^6)^{1/7}}{R0+\alpha},$$

$$\frac{\left(\cos\left(\frac{2\pi}{7}\right) - i\cos\left(\frac{3\pi}{14}\right)\right) \left((R0-1)(R0+\alpha)^6\right)^{1/7}}{R0+\alpha}$$

$$x1 := \left(\frac{R0-1}{R0+\alpha}\right)^{1/7} \quad (4)$$

> `y1:=factor(solve(B1,y1));`

$$y1 := \frac{\left(\frac{R0-1}{R0+\alpha}\right)^{1/7} R0(\alpha+1)}{7(R0+\alpha)(R0-1)} \quad (5)$$

> `x2:=solve(A2,x2);`

$$x2 := -\frac{4R0(\alpha+1)}{7(R0+\alpha)(R0-1)} \quad (6)$$

> `z1:=factor(solve(C1,z1));`

$$z1 := -\frac{\left(\frac{R0-1}{R0+\alpha}\right)^{1/7} R0(\alpha+1)(7R0^2 - R0\alpha - R0 + 7\alpha)}{49(R0+\alpha)^2(R0-1)^2} \quad (7)$$

> `y2:=solve(B2,y2);`

$$y2 := \frac{4(R0^2 + \alpha)R0(\alpha+1)}{7(R0+\alpha)^2(R0-1)^2} \quad (8)$$

> `x3:=factor(solve(A3,x3));`

$$x3 := -\frac{2R0(\alpha+1)\left(\frac{R0-1}{R0+\alpha}\right)^{6/7}(28R0^2 + 3R0\alpha + 3R0 + 28\alpha)}{49(R0+\alpha)(R0-1)^3} \quad (9)$$

Summarize the results for s=7:

> `x1, x2, x3, y1, y2, z1;`

$$\left(\frac{R0-1}{R0+\alpha}\right)^{1/7}, -\frac{4R0(\alpha+1)}{7(R0+\alpha)(R0-1)},$$

$$-\frac{2R0(\alpha+1)\left(\frac{R0-1}{R0+\alpha}\right)^{6/7}(28R0^2 + 3R0\alpha + 3R0 + 28\alpha)}{49(R0+\alpha)(R0-1)^3},$$

$$\frac{\left(\frac{R0-1}{R0+\alpha}\right)^{1/7} R0(\alpha+1)}{7(R0+\alpha)(R0-1)}, \frac{4(R0^2 + \alpha)R0(\alpha+1)}{7(R0+\alpha)^2(R0-1)^2},$$

$$-\frac{\left(\frac{R0-1}{R0+\alpha}\right)^{1/7} R0(\alpha+1)(7R0^2 - R0\alpha - R0 + 7\alpha)}{49(R0+\alpha)^2(R0-1)^2}$$

>